

QuESTlink – Mathematica embiggened by a hardware-optimised quantum emulator

Tyson Jones^{1,2,*} and Simon Benjamin^{1,†}

¹*Department of Materials, University of Oxford, Parks Road, Oxford OX1 3PH, United Kingdom*

²*with additional support from Quantum Motion Technologies Ltd UK*

(Dated: October 11, 2021)

We introduce **QuESTlink**, pronounced ‘quest link’, an open-source Mathematica package which efficiently emulates quantum computers. By integrating with the Quantum Exact Simulation Toolkit (QuEST), QuESTlink offers a high-level, expressive and usable interface to a high-performance, hardware-accelerated emulator. Requiring no installation, QuESTlink streamlines the powerful analysis capabilities of Mathematica into the study of quantum systems, even utilising remote multi-core and GPU hardware. We demonstrate the use of QuESTlink to concisely and efficiently simulate several quantum algorithms, and present some comparative benchmarking against core QuEST.

CONTENTS

I	Foreward	1
II	Introduction	1
	A. QuESTlink facilities	2
III	Technical summary	2
	A. Architecture	2
	B. Deployment	3
	C. Remote computation	3
IV	User guide	4
	A. Mathematica review	4
	B. QuESTlink overview	4
V	Demonstrations	6
	A. Decoherence	6
	B. Variational imaginary-time	6
	C. Noisy Trotterisation	7
	D. Further examples	9
VI	Benchmarking	9
VII	Future work	10
VIII	Conclusion	11
IX	Acknowledgements	11
	References	11

I. FOREWARD

This manuscript uses the phrases “simulation” and “emulation” interchangeably, to refer to any use of a classical computer to study, mimic or approximate the behaviour of a digital quantum computer. While QuESTlink can be classed as a *strong simulator* [1], we use the phrase emulation to distinguish from “quantum simulation”, that is the use of a quantum computer to study a quantum system.

II. INTRODUCTION

Classical emulation is crucial in the design of quantum computers and algorithms. Despite the recent demonstration of quantum supremacy [2], today’s quantum computers are of insufficient quality to run and test many interesting algorithms. Even precise quantum computers of tomorrow may provide limited help in writing new algorithms, since unlike emulators, they offer limited information about the evolving quantum state. Furthermore, some algorithms, particularly those for noisy intermediate-scale quantum (NISQ) devices [3] like the variational class of algorithms [4], admit limited analytic treatment. Hence, the value of classical emulation is undeniable.

The research community needs high-level usable tools that are easy to deploy, offer rapid numerical study, and integrate with other established software. However, the exponentially-growing cost of classically simulating a quantum device makes emulation of even NISQ computers very resource intensive. Emulators must therefore make good use of classical high-performance computing techniques, like multithreading and GPU parallelisation, and be written in low-level performant languages like C. This requirement is at odds with the need for usable tools, which can be used by non-expert programmers and the wider quantum community.

Within this context we have developed QuESTlink: a high-performance Mathematica package for numerically emulating quantum computers, by off-loading expensive computation to remote accelerated hardware, running QuEST [5]. Mathematica is both a language and computational tool, prevalent among physicists, which offers a convenient interactive interface (through notebooks), and an extremely comprehensive and powerful set of utilities. Although the most widely used tool for calculations in the physical sciences [6, 7], Mathematica does not have an intrinsic toolset specifically dedicated to quantum computing emulation. QuESTlink provides a high-performance emulator with a usable Mathematica interface, without compromising the excellent performance and simulation capacity of QuEST.

From a laptop environment, a user can symbolically

* tyson.jones@materials.ox.ac.uk

† simon.benjamin@materials.ox.ac.uk

specify a circuit in an intuitive high-level operator format akin to how they appear in the literature. Behind a platform-agnostic Mathematica interface, QuESTlink sends the circuit to a C++ backend, either locally or on remote high-performance hardware, where it is efficiently emulated. QuESTlink offers multithreaded and GPU simulation of state vectors and density matrices, multi-qubit multi-controlled general unitaries, general noise processes, circuit drawing, and a wide range of standard gates. Like QuEST, QuESTlink is free, open-source and can be used stand-alone; furthermore, without setup, compilation or installation of any kind.

A. QuESTlink facilities

While the forthcoming sections give a thorough overview of QuESTlink’s facilities, we here provide a short summary to acquaint the reader with the essentials. QuESTlink is, offers, features or enables:

- Multithreaded and GPU-accelerated emulation of quantum computers.
- State-vector *and* density matrix simulation.
- Off-loading of simulation to remote hardware, through a backend-agnostic interface.
- Seamless integration with Mathematica’s powerful and comprehensive toolset, and interactive notebook programming style.
- Rapid development with Mathematica’s concise, functional language.
- A concise but expressive circuit language, akin to the symbolic description of circuits in the literature.
- Stand-alone with no required external downloading, compiling, or installation of any kind whatsoever.
- Through Mathematica, is compatible with all major operating systems.
- Free and open-source, hosted publicly on Github [8].
- Rendering of circuit diagrams, which can be exported to any file format through Mathematica.
- A suite of functions for higher-level calculations, like computing Hamiltonian expectation values, and derivatives of unitary circuits.
- Analytic generation of matrices from circuit specifications, which can include symbols.
- A comprehensive suite of unitary gates and decoherence channels, including multi-controlled multi-qubit general unitaries, and multi-qubit Kraus maps.
- A rigorous user-input validation scheme to catch user errors, like supplying non-unitary matrices or physically invalid noise parameters.




FIG. 1. The QuESTlink software stack, from the user’s code (`user.nb`) to the driving QuEST [5] framework. This stack can run entirely on a user’s local machine, achieving an interactive Mathematica interface to a C/C++ based emulator. Note the distinction between WSTP (the protocol for communicating between Mathematica and an external process) and `C/Link` (an implementation for C) is unimportant, and hereafter disregarded.

III. TECHNICAL SUMMARY

This section provides an overview of the inner workings of QuESTlink, and the technologies used to build it. Readers intending to use QuESTlink right away may wish to skip to Section IV B.

A. Architecture

QuESTlink consists of both a Mathematica package (`QuESTlink.m`) and an underlying C++ program (`quest_link.cpp`), which interface using `C/Link`; an implementation of the Wolfram Symbolic Transfer Protocol (WSTP) [9] (formerly called `MathLink` [10]). QuESTlink emulates quantum computers using the Quantum Exact Simulation Toolkit (QuEST) [5], which is an open-source, high-performance emulator written in C. QuEST is multi-threaded, GPU-accelerated and distributed, and the first two of these facilities are made available to QuESTlink.

`C/Link` facilitates conversion of Mathematica types, utilised in the user’s notebook, into C types, and maps Mathematica-callable functions to C functions. The executable `quest_link` (`quest_link.c` compiled with WSTP) sits below as a “backend” process, performing intermediate processing and invoking QuEST’s API. This software stack is visualised in Figure 1.

Through this stack, QuESTlink offers a high-level Mathematica interface to quantum emulation which is numerically performed in C/C++ (a significantly faster but lower level language than Mathematica), using memory persistent in the C process, and potentially using accelerated hardware. The expensive numerical representations of the quantum states reside only in the C backend (and through QuEST, may also be persistent also in GPU

memory), and each are identified by a unique ID. Quantum circuits are represented symbolically in their entirety in Mathematica, and sent to the C backend (at a small runtime overhead) only at emulation-time, compactly encoded into arrays of real numbers. The circuit language, and details of the QuESTlink interface, are presented in Section IV B.

QuESTlink leverages QuEST’s extensive user-input validation, so that user errors are detected early and reported. For example, this includes validation that input matrices are unitary, that a gate’s control and target qubits are unique, and that parameters of a decoherence channel result in a completely positive map. As of QuESTlink v0.3 (integrating QuEST v3.1.0), there are 72 unique forms of input validation. This helps users avoid logical errors in their code, to mitigate the risk of drawing incorrect conclusions from the results of QuESTlink.

Core QuEST performs validation before any modification to a quantum register. If a problem is detected, a C++ exception is thrown and caught by QuESTlink, and a descriptive error message is propagated to the Mathematica kernel. Some QuESTlink validation is performed *on the fly*. For example, a problem in a gate may only be detected after applying the prior gates in a circuit. In these instances, QuESTlink always reverts the affected register to its original state before the circuit was applied. This means no register is left in an uncertain state.

B. Deployment

QuESTlink can be obtained and deployed entirely within a Mathematica notebook, without any installation or configuration. This is done by hosting the package code and `quest_link` executable(s) online in a [Github](#) repository, which the `qtechtheory.org` server redirects to. Figure 2 presents the process of serving the package to a user’s Mathematica kernel.

The `quest_link` executables for Windows, MacOS and Linux are automatically recompiled on the Github repository with every new release, using Github’s integrated continuous-delivery service (*Github Actions* [11]). This is for both security, ease of development, and its facility to push bug-fixes and updates seamlessly to the end user.

C. Remote computation

QuESTlink even enables quantum emulation using *remote* computing resources. Through WSTP, the `quest_link` process can run on a remote machine (e.g. a supercomputer) and communicate with the user’s local Mathematica kernel via TCP/IP. The remote machine can employ more powerful hardware than available on the user’s machine, and potentially emulate larger quantum states than can fit in the user’s local memory. The protocol of off-loading a circuit emulation to remote hardware is outlined in Figure 3. The tools and documentation for




FIG. 2. The protocol for stand-alone deployment of QuESTlink. First, a user calls Mathematica’s `Import[url]` function to fetch a copy of the `QuESTlink.m` package, hosted on [Github](#), but redirected from

[qtechtheory.org](#). This provides the QuEST’ namespace, and defines functions to connect to a `quest_link` backend; this backend can exist on the calling machine, or remotely as here pictured. More details on so called “QuEST environments” are presented in detail in the proceeding sections. Note the featured code snippets are simplified here for clarity, and their syntax is reviewed in Section IV A.




FIG. 3. Protocol for off-loading an emulation task onto a remote QuEST environment. Simulation structures declared in Mathematica are actually allocated remotely, possibly even in accelerated hardware memory. Quantum circuits, by first being encoded into arrays of numbers, are communicated to the server over the internet (via TCP/IP). Note the syntax used in the code snippets in this diagram are reviewed in Section IV A

setting up a remote QuESTlink server are provided between the Github repo [8], and [questlink.qtechtheory.org](#).

In this remote configuration, Mathematica calls to QuESTlink functions will involve network communication with the remote environment, and hence incur overheads; this is worthwhile if the remote hardware sufficiently accelerates an expensive emulation which otherwise dominates runtime.

Despite QuESTlink’s effort to minimise the communication cost, this network overhead will be prohibitively costly for some applications. For example, when large simulated quantum states undergo processing by the Mathematica kernel, and hence need to be copied back and forth between the local kernel and the remote environment. To mitigate this slowdown whilst still running QuESTlink remotely, one can launch their Mathematica

kernel on the remote machine, using Mathematica’s remote kernel facilities [12]. QuESTlink can also be used inside other Mathematica packages, and so be launched remotely and non-interactively, without a local notebook.

IV. USER GUIDE

A. Mathematica review

Before continuing, we offer a quick review of the Mathematica syntax used in this manuscript.

Evaluation of function f with input a is denoted by $f[a]$, or equivalently $f @ a$. The expression $g[f[a]]$ can be formed using prefix notation as $g @ f @ a$, or postfix notation as $a // f // g$. Matrix multiplication between (possibly complex) matrices a and b is denoted by $a.b$, and a^\dagger denotes the conjugate transpose of a . Expressions with a trailing `;` suppress their otherwise displayed result. While $f[] = a$ denotes immediate evaluation of a and assignment to $f[]$, the syntax $f[] := a$ denotes *delayed* assignment, whereby later invoking $f[]$ will evaluate a (which may have changed) each time. Elements of a list $x = \{a, b, c\}$ are accessed as $x[[i]]$ where index $i \geq 1$, and $x[[m; n]]$ returns the sublist spanning indices m to n . The shortcut $ex /. a \rightarrow b$ replaces sub-expressions of ex which match pattern a , with b . Comments appear in parentheses (** like this **).

Finally, for those copying code into Mathematica, subscripts can be quickly entered into a notebook with keyboard shortcut `Ctrl` & `-`.

In the code snippets featured in this manuscript, context should make clear what is *input* to the Mathematica notebook, and what is a rendered *output*.

B. QuESTlink overview

The QuESTlink package requires no installation, and can be downloaded directly from within Mathematica:

```
Import["https://qtechtheory.org/QuESTlink.m"]
```

The user then has a choice of several “QuEST environments” in which to perform quantum emulation, all of which provide an identical user experience.

```
CreateDownloadedQuESTEnv[]
CreateLocalQuESTEnv[fn]
CreateRemoteQuESTEnv[ip, port1, port2]
```

The first, `CreateDownloadedQuESTEnv[]`, enables simulation on the user’s machine by downloading a serial QuESTlink executable, pre-compiled for the user’s operating system. This requires no apriori setup whatsoever.

`CreateLocalQuESTEnv[fn]` will attempt to launch an existing local QuESTlink executable, located at `fn`. This allows local simulation using serial, multi-core or GPU resources, depending on how the executable was compiled. Users can compile QuESTlink for their platform using the tools on the Github repo [8].

`CreateRemoteQuESTEnv[ip, port1, port2]` connects to a remote QuESTlink environment at the given `ip` address and ports. The remote machine can use serial, multithreading or GPU-acceleration to emulate quantum systems. The facilities to setup a remote QuESTlink server are provided in the Github repo [8].

Once connected to a QuEST environment, a full list of the supported QuESTlink facilities and gate symbols can be obtained by evaluating

```
?QuEST`*
```

and the documentation of a particular function or operator obtained similarly using `?`.

```
?M
```

M is a destructive measurement gate which measures the indicated qubits in the Z basis.

```
?CalcHilbertSchmidtDistance
```

CalcHilbertSchmidtDistance[qureg1, qureg2] returns the Hilbert–Schmidt distance (Frobenius norm of the difference) between the given density matrices.

Emulation begins by creating quantum registers (a “Qureg”), each represented by a state-vector or density-matrix.

```
numQb = 6;
ψ = CreateQureg[numQb];
ρ = CreateDensityQureg[numQb];
```

These functions return a unique ID for each Qureg, the memory for which is stored in the QuEST environment. Once created, QuESTlink provides a few functions to initialise Quregs:

```
InitZeroState[ψ]      (* |ψ⟩ → |0⟩⊗ *)
InitPlusState[ψ]      (* |ψ⟩ → |+⟩⊗ *)
InitClassicalState[ψ, 3] (* |ψ⟩ → |0⟩⊗ |11⟩ *)
InitPureState[ρ, ψ]   (* ρ → |ψ⟩⟨ψ| *)
```

or directly modify them (here, creating a random density matrix):

```
x = Table[RandomComplex[], 2numQb, 2numQb];
m = x . x† / Tr[x . x†];
SetQuregMatrix[ρ, m];
```

A quantum circuit u can be applied to a **Qureg**, agnostic of whether it is a state vector (effecting $u|\psi\rangle$) or a density matrix (effecting $u\rho u^\dagger$), using `ApplyCircuit[u, qureg]`. QuESTlink features a concise and expressive language for specifying gates and decoherence processes, where target qubits are denoted with subscript integers to gate symbols, and control gates “wrap” the base gate. The below example makes use of the `Circuit` function, which disables commutation and allows a concise product representation of the circuit. In combination, this enables a syntax akin to how circuits are denoted in the quantum computing literature.

```
m1[θ_] := ( 0      i
             eiθ/3 0 );
m2[] := RandomVariate @
  CircularUnitaryMatrixDistribution[4];

u[θ_] := Circuit[
  X0 Y1 Z2 T3 S4 H5 Rx0[θ] Ry1[θ] Rz2[θ] Rz3,5[θ] ×
  C0[X1] C4[Ry2[θ]] M5 R[θ, X0 Y3 X1] P4,5[0, 0] ×
  C2[U1,3[m2[]]] U0[m1[θ]] SWAP4,5 C2,3[SWAP0,1] ×
  Damp4[θ/102] Depol5[θ/102] Deph1,2[θ/102] ×
  Kraus0[{m1[θ]}] Kraus3,4[{m2[]}];


ApplyCircuit[u[π/3], ρ];
```

The result of `Circuit[]` is just a list of operators, allowing easy circuit manipulation. Mathematica features many tools for such lists allowing the user to easily extend, alter, join, compare, etc, their quantum circuit. For example:

```
v = Circuit[X2 H0 Y3];
v[[2]] = T0;
v
{X2, T0, Y3}

v /. Y → S
{X2, T0, S3}
```

These circuits can be swiftly visualised with `DrawCircuit[u]`, which supports saving rendered circuit diagrams to raster and vector images.



After applying a unitary circuit, one can of course opt to obtain the entire state vector or density matrix, as for example:

```
GetQuregMatrix[ρ] // MatrixForm
( 0.0465478    0. + 0.139643 i    -0.0794974    ...
  0. - 0.139643 i    0.41893    0. + 0.238492 i    ...
  -0.0794974    0. - 0.238492 i    0.139643    ...
  ⋮                ⋮                ⋮                ⋮ )
```

However, it is typically more efficient to calculate desired quantities, like the expectation value of a Hamiltonian in the Pauli-basis, in the QuEST environment itself.

```
h = .1 X1 Y0 Z3 + X5 X4 - .1 Z1 Z5;
φ = CreateDensityQureg[6];
CalcExpecPauliSum[ρ, h, φ]
0.0730679
```

At any time, **Quregs** can be individually (or, all-together) destroyed to free up memory in the QuEST environment.

```
DestroyQureg[ρ]
DestroyAllQuregs[]
```

By leveraging Mathematica’s powerful suite of symbolic calculations, QuESTlink can even build analytic matrices from circuit specifications.

```
Circuit[
  H0 C0[X1] C1[Ry0[φ]] C0[U1 @ ( α β
                                           γ δ )];
CalcCircuitMatrix[%] // MatrixForm
( 1/√2      1/√2      0      0
  β Cos[φ/2]/√2 - β Cos[φ/2]/√2 - α/√2 + β Sin[φ/2]/√2 - α/√2 + β Sin[φ/2]/√2
  - Sin[φ/2]/√2 - Sin[φ/2]/√2 - Cos[φ/2]/√2 - Cos[φ/2]/√2
  δ Cos[φ/2]/√2 - δ Cos[φ/2]/√2 - γ/√2 + δ Sin[φ/2]/√2 - γ/√2 + δ Sin[φ/2]/√2 )
```

With these facilities, QuESTlink offers a seamless integration with Mathematica’s comprehensive range of com-

putational and graphical tools, as illustrated in the following demonstrations.

V. DEMONSTRATIONS

To demonstrate the concision [13] offered by QuESTlink, we provide several examples of somewhat sophisticated computations written in only several lines, and efficiently simulated. For users wishing to run these demonstrations directly, they are compiled into a single notebook at questlink.qtechtheory.org/paper_demos.nb.


A. Decoherence

To begin, we very compactly demonstrate the effect of two-qubit depolarising noise on the expected measurement of a simple Hamiltonian h . Starting in a random pure state ψ , a depolarising channel with probability 0.1 of any Pauli error occurring is repeatedly applied, in total 100 times.

```
(* create a random initial state *)
{ψ, φ} = CreateQuregs[5, 2];
{ρ, σ} = CreateDensityQuregs[5, 2];
SetQuregMatrix[ψ, Normalize @
  Table[RandomComplex[], 25]];
InitPureState[ρ, ψ];

(* calculate E[h] while depolarising *)
h = .3 + .1 X0 Y1 Z2 - .2 Z0;
data = Table[
  MixTwoQubitDepolarising[ρ, 0, 1, .1];
  {CalcFidelity[ρ, ψ],
   CalcExpecPauliSum[ψ, h, φ] -
   CalcExpecPauliSum[ρ, h, σ]},
  100];

ListLinePlot[Transpose[data], opts]
```



Note that undisclosed variable `opts` contained addi-

tional code for customising the plot.

B. Variational imaginary-time

In this demonstration, we emulate the quantum variational imaginary-time simulation routine [14] to approximate the ground-state of a molecular Hamiltonian.

We first download a reduced 6-qubit representation of the electronic structure Hamiltonian of Lithium Hydride (LiH).

```
nQb = 6;
h = GetPauliSumFromCoeffs[
  "https://qtechtheory.org/hamil_6qbLiH.txt"]

h[-2 ;;]
0.0591748 Z1 Z2 Z4 Z5 + 0.147366 Z3 Z4 Z5
```


We create a simple 6-qubit ansatz circuit featuring 39 parameters, denoted with variables $\vec{\theta} = \{\theta_1, \dots, \theta_{39}\}$.

```
i = 1;
entangle[qbs_] :=
  Table[R[θ[i++], σq σq+1], {σ, {X, Y, Z}}, {q, qbs}]

(* create {Rz0[θ[1]], Rz1[θ[2]], Rz2[θ[3]], ...} *)
ansatz = Flatten @ Join[
  (* single-qubit rotations on every qubit *)
  Table[rq-1[θ[i++]], {r, {Rz, Ry, Rx, Rz}}, {q, nQb}],
  (* two-qubit rotations neighbouring pairs *)
  entangle[{0, 2, 4}],
  entangle[{1, 3}]];

nθ = Length[ansatz]
```

This `ansatz` circuit consists of one and two qubit rotation gates; $\exp(-i\theta_j \hat{\sigma}/2)$, for $\hat{\sigma} \in \{X, Y, Z, X \otimes X, Y \otimes Y, Z \otimes Z\}$, which in QuESTlink are denoted by $Rx_q[\theta]$ (etc.) and $R[\theta, X_{q1}X_{q2}]$. In the diagram below, each such paired two-qubit rotation is marked by a vertical link.



We now create several `Quregs` with which to emulate the quantum algorithm.

```
{ψ, hψ, φ} = CreateQuregs[nQb, 3];
dψ = CreateQuregs[nQb, nθ];
```

ψ will be maintained as the output state of the ansatz circuit, and $h\psi$ will store the result of applying the Hamiltonian h to ψ . ϕ will merely provide intermediate work-space for calculations, and $d\psi$ will store a Qureg for each parameter in the ansatz (a total of $n\theta$).

Next, we choose a random initial assignment of the ansatz parameters, and measure the energy of the resulting quantum state.

```
cure = Table[θt → RandomReal[], {t, nθ}];
ApplyCircuit[ansatz /. cure, ψ];
CalcExpecPauliSum[ψ, h, φ]
-5.87043
```

The variational imaginary-time algorithm [14] involves repeatedly measuring a matrix and vector of quantities,

$$A_{ij} = \text{Re} \left\{ \left\langle \frac{\partial \psi(\vec{\theta})}{\partial \theta_i} \middle| \frac{\partial \psi(\vec{\theta})}{\partial \theta_j} \right\rangle \right\}, \quad (1)$$

$$C_i = -\text{Re} \left\{ \left\langle \psi(\vec{\theta}) \middle| h \middle| \frac{\partial \psi(\vec{\theta})}{\partial \theta_i} \right\rangle \right\} \quad (2)$$

and iteratively updating the parameters under

$$\vec{\theta} \rightarrow \vec{\theta} + \Delta t A^{-1} \vec{C}, \quad (3)$$

which we now concisely emulate for nt iterations.

```
Δt = .01;
nt = 100;
Do[
  InitZeroState[ψ];

  (* compute |∂Uψ/∂θj⟩ for all j *)
  CalcQuregDerivs[ansatz, ψ, cure, dψ];
  matrA = CalcInnerProducts[dψ] // Re;

  (* compute ⟨ψ|h|∂Uψ/∂θj⟩ *)
  ApplyCircuit[ansatz /. cure, ψ];
  ApplyPauliSum[ψ, h, hψ];
  vecC = -CalcInnerProducts[hψ, dψ] // Re;

  (* update θ → θ + Δt A-1C *)
  Δθ = Δt LinearSolve[matrA, vecC];
  cure[[All, 2]] += Δθ,
  nt
]
```

The energy of the output quantum state, as produced by the reached assignment of the parameters, is close to

the true groundstate found through matrix diagonalisation.

```
CalcExpecPauliSum[ψ, h, φ]
-7.87037

Min @ Eigenvalues @ CalcPauliSumMatrix @ h
-7.88074
```

Interestingly, a significantly slower but direct minimisation in Mathematica reveals the energy reached by imaginary-time was *not* quite the best possible of our chosen ansatz.

```
(* E[ψ] = ⟨0|U†hU|0⟩, where U=U(θ1,...) *)
energy[θvals_?NumericQ] := Module[{cure},
  cure = Table[θ[t] → {θvals}[[t]], {t, nθ}];
  InitZeroState[ψ];
  ApplyCircuit[ansatz /. cure, ψ];
  CalcExpecPauliSum[ψ, h, φ]

  θvars = Table[θt, {t, nθ}];
  NMinimize[energy @@ θvars, θvars][[1]]
-7.87954
```

We invite the interested reader to compare the Mathematica code above to a native C implementation of imaginary time evolution, as hosted on Github [15].

C. Noisy Trotterisation

In this demonstration, we emulate Trotterisation of a spin-ring Hamiltonian, with and without noise, and compare it to direct numerical solving of the Schrödinger equation. Our Hamiltonian is the one-dimensional nearest-neighbour (periodic boundary conditions) nQb -spin Heisenberg model with a random magnetic field in the z direction,

$$\sum_j \vec{\sigma}_j \vec{\sigma}_{j+1} + r_j \sigma_j^z, \quad r_j \sim \mathcal{U}[-1, 1]. \quad (4)$$

Real-time simulation of this model to time $t = nQb$ is nominated by Childs *et al.* [16] as an early practical application of a quantum computer. In this example, we'll study a five-spin chain Hamiltonian h .

```
nQb = 5;
h = Flatten @ Join[
  Table[1. σq-1 σMod[q,nQb], {σ, {X, Y, Z}}, {q, nQb}],
  Table[RandomReal[{-1, 1}] Zq-1, {q, nQb}]]
```

To simulate evolution on a quantum computer, we will emulate circuits formed by the Suzuki-Trotter decompo-

sitions [17] of the unitary evolution operator of varying order n . Below, r is the number of repetitions of the order- n circuit to perform, to ultimately reach time t .

```

symmetrize[h_, λ_, 1] := λ h
symmetrize[h_, λ_, 2] := With[
  {s1 = symmetrize[h, λ/2, 1]},
  Join[s1, Reverse[s1]]]
symmetrize[h_, λ_, n_?EvenQ] := Block[
  {γ, p = 1/(4 - 41/(n-1))}, With[
    {s = symmetrize[h, γ, n-2]}, With[
      {r = s /. γ → λ p},
      Join[r, r, s /. γ → (1-4 p) λ, r, r]]]]
gateify[Verbatim[Times][θ_, σ_]] :=
  R[2 θ, Times[σ]]
trotterize[h_, n_, r_, t_] := With[
  {s = symmetrize[h, t/r, n]},
  gateify /@ Flatten @ ConstantArray[s, r]]

```

Note the original ordering of the Hamiltonian terms is arbitrary, and should ideally be optimised into commuting groups, or at least randomly shuffled. In particular, Childs *et al.*'s uniform randomisation scheme [18], whereby each repetition sees a random Hamiltonian ordering, is trivially implemented as:

```

childsify[h_, n_, r_, t_] := Flatten @ Table[
  symmetrize[RandomSample @ h, t/r, n], r]

```

Even Campbell's qDRIFT routine [19] can be given a compact implementation.


```

campbellize[h_, r_, t_] := With[
  {c = h[[All, 1]], σ = h[[All, 2 ;;]], N = Length[h] * r},
  With[{λ = Total[c], p = Abs @ Normalize[c, Total]},
    t * λ / N RandomChoice[p → σ, N]]]

```

However, we opt instead to utilise only deterministic Trotterization for clarity. For example, the first-order ($n=1$) single-repetition ($r=1$) Trotter circuit (arbitrarily to time $t=1$) has the form:

```
DrawCircuit @ trotterize[h, 1, 1, nQb]
```



We compare the states produced from Trotter circuits with the “true” (to numerical precision) time evolution, found by numerically solving the Schrödinger equation

using Mathematica's in-built `NDSolveValue` routine.

```

schrod[h_, ψ0_, t_] := With[
  {H = CalcPauliSumMatrix[Total@h]},
  NDSolveValue[
    {I ψ'[τ] == H.ψ[τ], ψ[0] == ψ0}, ψ,
    {τ, 0, nQb}]] [t]

```

In the proceeding solutions, the Qureg $\psi0$ will store the initial state (a random pure state), ψt will store the true state, and ψ will store outputs of the Trotter circuits.

```

ψ0v = Normalize @ Table[RandomComplex[], 2nQb];
ψ0 = CreateQureg[nQb];
SetQuregMatrix[ψ0, ψ0v];

ψt = CreateQureg[nQb];
SetQuregMatrix[ψt, schrod[h, ψ0v, nQb]];

ψ = CreateQureg[nQb];
CloneQureg[ψ, ψ0];

```

We can now commence the computation by generating circuits of a varying Trotter order and number of repetitions, emulating their execution, and computing the fidelity of their output state with the true state.


```

fids = Table[
  circ = trotterize[h, order, reps, nQb];
  ApplyCircuit[circ, ψ0, ψ];
  {Length[circ], CalcFidelity[ψ, ψt]},
  {order, {1, 2, 4, 6, 8}},
  {reps, 1, 50}
];

```

We then plot the result below, using some undisclosed parameters `opts` to tweak the plot style. The horizontal axis is the total number of gates in the Trotter circuit.

```
ListLogLinearPlot[fids, opts]
```




We can even imitate a noisy quantum device by inserting decoherence operators into our circuit. In this

demonstration, we will follow each Trotter-prescribed gate with one or two qubit depolarising operators, which effect the channel

$$\text{Depol}[\rho] = (1 - p)\rho + \frac{p}{3} \sum_{\sigma} \sigma \rho \sigma, \quad (5)$$

(noting $\sigma = \sigma^\dagger$ for Paulis) and similarly for the two-qubit analogue. We choose an error rate of $p = 10^{-4}$, below the rates recently demonstrated in Google's Sycamore machine [2]. Below is a visualisation of the first five unitary gates of the resulting noisy circuit.



Our simulation of the noisy circuit is near-identical to our simulation of pure states, with the exception that we now operate upon density matrices.

```
{ρ, ρ0} = CreateDensityQuregs[nQb, 2];
InitPureState[ρ0, ψ0];

nfids = Table[
  circ = noisify[10-4] @
    trotterize[h, order, reps, nQb];
  ApplyCircuit[circ, ρ0, ρ];
  {Length[circ], CalcFidelity[ρ, ψt]},
  {order, {1, 2, 4, 6, 8}},
  {reps, 1, 50}
];
```

As one might expect, the accuracy of the Trotter circuits have waned, and eventually decrease with increased circuit depth, due to the opportunity for additional errors to accrue.





FIG. 4. A 5-qubit 2-repetition example of the arbitrary circuit used for comparative benchmarking of QuESTlink and QuEST. Every gate involves a rotation of a uniformly random angle between 0 and 4π , which generate all rotations. The 15-qubit circuits employed for benchmarking feature 87 gates per repetition.



D. Further examples

Complete notebooks demonstrating some extended computations in QuESTlink are available at questlink.qtechtheory.org.

VI. BENCHMARKING

We now benchmark QuESTlink's local and remote emulation of some arbitrary quantum circuits, and compare to benchmarks of direct simulation in C, using QuEST. This quantifies the runtime overheads incurred by Mathematica integration, and whether QuESTlink is the right emulation tool for the user's target simulation regime.

We nominate a simple circuit consisting of X , Y , and Z axis rotations of random angles on each qubit, followed by tessellated controlled rotations on every pair of neighbouring qubits. This pattern is repeated in the circuit, and is illustrated in Figure 4. These benchmarks will emulate 15-qubit circuits, with between 1 to 50 repetitions (an upper bound of 4350 gates), and each will be simulated 10 times with re-randomised angles.

Core QuEST is profiled directly in C, through precise timing of each full circuit execution. In contrast, QuESTlink is profiled through a notebook using Mathe-




FIG. 5. Comparative performance of QuEST and (local) QuESTlink emulating the 15-qubit circuit illustrated in Figure 4, with a varying circuit depth (repetitions of the base circuit), in each of QuESTlink’s supported parallelisation modes. Multithreaded emulation used 8 of 12 available threads, though were shared with the Mathematica kernel. Solid lines indicate the mean runtime of 10 random simulations, and shaded regions indicate three standard deviations. The circuit of 50 repetitions contains a total of 4350 gates.

Mathematica’s `AbsoluteTiming[]` function; its runtime will include Mathematica evaluation, `QuESTlink.m` preprocessing and circuit encoding, the `C/Link` and WSTP overheads, `quest_link.c`’s decoding of the circuit, and ultimately QuEST’s simulation.

Benchmarking is performed on a 12-core Xeon W-2133 3.6 GHz CPU. At most 8 threads will be employed in multithreaded mode, so as not to interfere with threads used by the Mathematica kernel. GPU-accelerated testing will employ a 24 GB NVIDIA Quadro P6000 in the same machine.

The results of benchmarking are presented in Figure 5, and are as expected. The Mathematica overhead of invoking serial QuEST through QuESTlink is small (a factor ≈ 1.1). This manifests from a fixed per-gate cost of encoding the circuit, though its proportion of the total runtime will depend on the total runtime, and hence the number of simulated qubits.

Multithreading with 8-threads offered a $\approx 5\times$ speedup but introduced additional variation in runtime, most likely due to dynamic reallocation of threads between the Mathematica kernel and the backend QuEST process. Otherwise, the multithreading overhead is similarly small (a factor ≈ 1.08 for these tests). This reveals that the Mathematica pre-processing itself is accelerated by multithreading with the remaining four available threads.

At first glance, GPU QuESTlink may appear anomalously slow; on average 7.2 times slower than core GPU QuEST. However, this is due to QuEST simulation being very rapid, yielding a proportionately larger Mathematica overhead. For example, in the largest serial simulation, QuESTlink was on average 5×10^{-2} seconds slower than QuEST. This is comparable to the largest overhead of 4×10^{-2} in the GPU tests.

To confirm this, we performed additional GPU




FIG. 6. Benchmarking of a fixed-length circuit simulation in QuESTlink, using different environment configurations, though all ultimately running on the previously profiled Xeon machine’s GPU. *Remote* used `CreateRemoteQuESTEnv[]` to connect from a 13-inch 2017 MacBook Pro, with an average latency of 0.38 ms, with wall time measured in Mathematica with `AbsoluteTiming[]`. *Local* ran in a notebook on the Xeon machine, also using `AbsoluteTiming[]`. *Internal* measured the time spent in the QuESTlink C++ backend, and so excludes any Mathematica and WSTP overheads. Each datum is the average of 50 tests with re-randomised gate parameters.

benchmarking with a fixed circuit of 100 random rotations, and varying number of qubits. We study the overheads of local and remote emulation, using both `CreateLocalQuESTEnv[]` and `CreateRemoteQuESTEnv[]`. The results are presented in Figure 6, and reveal a negligible overhead on all platforms.

In general, use of a *remote* QuEST environment will add a constant overhead to QuESTlink’s performance, due to network latency. Though in principle the cost of communicating a circuit from Mathematica to a remote backend scales linearly with the emulated circuit depth, this cost should be overshadowed by the exponentially growing cost of quantum emulation, and other network overheads. Indeed, this is exhibited in Figure 6. The size of an encoded circuit is upper bounded by 8Σ bytes, where Σ is the total number of gates, control qubits, target qubits and gate parameters (e.g. angles of rotation). A circuit would need to contain approximately 30 million single-qubit gates to saturate a 1 GB bandwidth network.

VII. FUTURE WORK

QuESTlink is currently under active development, with a growing list of planned work and new features. This list includes:

- Integration with a GPU-accelerated linear algebra library, to perform fast numerical routines directly on the simulated quantum state. For example, diagonalisation of a density matrix enables rapid calculation of the Trace distance, which currently requires expensive communication with the Mathematica kernel.

- Support for emulation on remote, *distributed* hardware.
- A QASM [20] parser and generator, to and from QuESTlink’s circuit specification language.

We caution that QuESTlink is still in an early form and likely to change, both in interface and architecture.

VIII. CONCLUSION

This manuscript introduced QuESTlink, a Mathematica package for emulating quantum circuits, state-vectors and density matrices. QuESTlink offers both high-level symbolic manipulation of quantum circuits, and rapid simulation using possibly remote hardware, such as multicore and GPU-accelerated supercomputers. We presented a broad technical overview of QuESTlink, including its protocol for stand-alone installation-free deployment. We then demonstrated the concision and flexibility possible of QuESTlink, by stepping through several examples of otherwise sophisticated simulations. These examples should enable an interested reader to begin using QuESTlink immediately. Lastly, we per-

formed some simple benchmarks of QuESTlink to estimate the overhead over core QuEST, across its parallelisation modes. QuESTlink is open-source, and accessible at questlink.qtechtheory.org, or on Github [8]

IX. ACKNOWLEDGEMENTS

The authors sincerely thank Balint Koczor for helping porting QuESTlink to Windows, and performing extensive testing of QuESTlink’s analytic capabilities. We also thank Suguru Endo and Balint Koczor for helping shape the functionality and direction of QuESTlink and bug-spotting. Thanks also goes to the early adopters of QuESTlink in the QTechTheory group, for their part in moulding a perfectly cromulent emulator. TJ additionally thanks Quantum Motion Technologies Ltd for financial support in extending QuESTlink, and Sinan Shi for his noble spirit and expertise in rubber duck debugging. SCB acknowledges EPSRC grants EP/M013243/1 and EP/T001062/1, the IARPA funded project LogiQ, and the EU Flagship project AQTION. The authors would like to acknowledge the use of the University of Oxford Advanced Research Computing facility (ARC, please see <http://dx.doi.org/10.5281/zenodo.22558>) in enabling the related QuEST work.

-
- [1] Cupjin Huang, Michael Newman, and Mario Szegedy. Explicit lower bounds on strong quantum simulation, 2018.
 - [2] Frank Arute, Kunal Arya, Ryan Babbush, Dave Bacon, Joseph C Bardin, Rami Barends, Rupak Biswas, Sergio Boixo, Fernando GSL Brandao, David A Buell, et al. Quantum supremacy using a programmable superconducting processor. *Nature*, 574(7779):505–510, 2019.
 - [3] John Preskill. Quantum Computing in the NISQ era and beyond. *Quantum*, 2:79, August 2018.
 - [4] Jarrod R McClean, Jonathan Romero, Ryan Babbush, and Alán Aspuru-Guzik. The theory of variational hybrid quantum-classical algorithms. *New Journal of Physics*, 18(2):023023, 2016.
 - [5] Tyson Jones, Anna Brown, Ian Bush, and Simon C Benjamin. QuEST and high performance simulation of quantum computers. *Scientific reports*, 9(1):1–11, 2019.
 - [6] Andrey Grozin. *Introduction to Mathematica® for Physicists*. Springer, 2014.
 - [7] Patrick T Tam. *A physicist’s guide to Mathematica*. Academic Press, 2011.
 - [8] Tyson Jones. QuESTlink (public Github repo). <https://github.com/QTechTheory/QuESTLink>, 2019.
 - [9] WSTP (Wolfram Symbolic Transfer Protocol). <https://www.wolfram.com/wstp/>. Accessed: 2019-12-05.
 - [10] Todd Gayley. A MathLink tutorial. *Wolfram Research*, 2002.
 - [11] Github Actions. <https://github.com/features/actions>. Accessed: 2020-03-15.
 - [12] How to — connect to a remote kernel, wolfram language & system, documentation center. <https://reference.wolfram.com/language/howto/ConnectToARemoteKernel.html>. Accessed: 2019-12-05.
 - [13] Brad Leithauser. In praise of concision. *The New Yorker*, Feb 2013.
 - [14] Sam McArdle, Tyson Jones, Suguru Endo, Ying Li, Simon C Benjamin, and Xiao Yuan. Variational ansatz-based quantum simulation of imaginary time evolution. *npj Quantum Information*, 5(1):1–6, 2019.
 - [15] Tyson Jones. Tiny Imaginary Time (public Github repo). <https://github.com/TysonRayJones/TinyImagTime>, 2019.
 - [16] Andrew M. Childs, Dmitri Maslov, Yunseong Nam, Neil J. Ross, and Yuan Su. Toward the first quantum simulation with quantum speedup. *Proceedings of the National Academy of Sciences*, 115(38):9456–9461, 2018.
 - [17] N Hatano and M Suzuki. Finding exponential product formulas of higher orders, 2005.
 - [18] Andrew M. Childs, Aaron Ostrander, and Yuan Su. Faster quantum simulation by randomization. *Quantum*, 3:182, September 2019.
 - [19] Earl Campbell. Random compiler for fast hamiltonian simulation. *Physical review letters*, 123(7):070503, 2019.
 - [20] Andrew W. Cross, Lev S. Bishop, John A. Smolin, and Jay M. Gambetta. Open quantum assembly language, 2017.